Biom-Sensor (Befehle)

Ein **Biom-Sensor** kann prüfen ob der Spieler sich in einem bestimmten Biom befindet. Die Prüfung kann durch den Befehl `/execute if biome`, wie auch durch Fortschritte oder Prädikate erfolgen. Man kann jedoch nicht an einer beliebigen Position in der Welt den Namen eines beliebigen Bioms erhalten.

## Variante 1

* Grundmodule
* Befehl
* Befehlssignal
* Grund-Element (Befehle)
* Signal-Element (Befehle)
* Signalleitung (Befehle)
 * Taktgeber (Befehle)
* JSON-Text
* NBT

* 1 Redstone-Fackel
* 2 Befehlsblock
* 10 Ketten-Befehlsblock
* 1 Wiederhol-Befehlsblock

Dieser Biom-Sensor kann nur das Biom die Leere erfassen. Wählt man den Stock aus, wird einem über der Schnellleiste dies angezeigt, falls man sich nicht in der Leere befindet, wird eine negierte Meldung ausgegeben.

Einschalten
* Befehl 1.1 Die Spieler erhalten einen magischen Stock, welchen man in der Schnellleiste auswählen kann, um zu prüfen ob man in dem Biom ist.
* Befehl 1.2 Der Chunk wird dauerhaft geladen, damit man sich auch weiter entfernen kann.

Wiederholen
* Befehl 1.3 Spieler die den Stock auswählen erhalten ein temporäres Etikett.
* Befehl 1.4 Wenn der Spieler den Stock ausgewählt hat, wird ihm über der Schnellleiste das Biom die Leere angezeigt, sofern er sich dort befindet, ansonsten wird diese Meldung negiert[B1.5].
* Befehl 1.6 Falls der Spieler den Stock gerade nicht mehr ausgewählt hat, wird ihm der Text über der Schnellleiste geleert.
* Befehl 1.7 Das Etikett wird vergeben oder entfernt[B1.8], wenn der Spieler im gleichen Augenblick den Stock entweder ausgewählt hat oder nicht mehr ausgewählt hat.
* Befehl 1.9 Das Etikett für das Auswählen des Stockes wird entfernt, damit es beim nächsten Durchlauf nur vorhanden ist, wenn der Stock ausgewählt wurde.

Ausschalten
* Befehl 1.10 Das eventuell noch vorhandene Etikett wird entfernt.
* Befehl 1.11 Der Stock wird aus allen Inventaren der Spieler entfernt.
* Befehl 1.12 Falls der Stock auf den Boden geworfen wurde, wird er entfernt.
* Befehl 1.13 Der Chunk wird entladen.

```
give @a[distance=..15] minecraft:stick{EigBiomS.1Alle:true,EigBiomS.1Sensor:true,Enchantments:[{id:-1}],display:{Name:'{"text":"Biom-Sensor","color":"dark_purple","bold":true}',Lore:['"Wähle den Stock aus"','"um nach dem Biom"','"zu testen."'] } }

```

↑ 1.2:

```
forceload add ~ ~

```

↑ 1.3:

```
tag @a[nbt={SelectedItem:{tag:{EigBiomS.1Sensor:true} } }] add EtiBiomS.1Ausgewaehlt

```

↑ 1.4:

```
execute as @a[tag=EtiBiomS.1Ausgewaehlt] at @s if biome ~ ~ ~ minecraft:the_void run title @s actionbar ["",{"text":"Biom == ","color":"gray","bold":true},{"text":"Die Leere","color":"blue","bold":true}]

```

↑ 1.5:

```
execute as @a[tag=EtiBiomS.1Ausgewaehlt] at @s unless biome ~ ~ ~ minecraft:the_void run title @s actionbar ["",{"text":"Biom == ","color":"gray","bold":true},{"text":"NICHT Die Leere","color":"red","bold":true}]

```

↑ 1.6:

```
title @a[tag=!EtiBiomS.1Ausgewaehlt,tag=EtiBiomS.1Auswahl] actionbar [""]

```

↑ 1.7:

```
tag @a[tag=EtiBiomS.1Ausgewaehlt,tag=!EtiBiomS.1Auswahl] add EtiBiomS.1Auswahl

```

↑ 1.8:

```
tag @a[tag=!EtiBiomS.1Ausgewaehlt,tag=EtiBiomS.1Auswahl] remove EtiBiomS.1Auswahl

```

↑ 1.9:

```
tag @a[tag=EtiBiomS.1Ausgewaehlt] remove EtiBiomS.1Ausgewaehlt

```

↑ 1.10:

```
tag @a[tag=EtiBiomS.1Auswahl] remove EtiBiomS.1Auswahl

```

↑ 1.11:

```
clear @a minecraft:stick{EigBiomS.1Alle:true}

```

↑ 1.12:

```
kill @e[type=minecraft:item,nbt={Item:{tag:{EigBiomS.1Alle:true} } }]

```

↑ 1.13:

```
forceload remove ~ ~

```

## Variante 2

* Grundmodule
* Befehl
* Befehlssignal
* Grund-Element (Befehle)
* Signal-Element (Befehle)
* Signalleitung (Befehle)
 * Datenpaket
* Fortschrittsdaten
* JSON-Text
* NBT

* 1 Redstone-Fackel
* 2 Befehlsblock
* 1 Wiederhol-Befehlsblock

Dieser Biom-Sensor erkennt mit Hilfe eines Fortschrittes, ob der Spieler im Biom die Leere sich befindet. Nur wenn er sich dort befindet, bekommt er eine Schnellleisten-Nachricht.

Einschalten
* Befehl 2.1 Auf dem Wiederhol-Befehlsblock befindet sich ein Hebel. Legt man ihn um, wird zuerst der Befehlsblock mit der Start-Funktion biom-sensor:v2start aktiviert.

Wiederholen
* Befehl 2.2 Gleichzeitig wird auch der Wiederhol-Befehlsblock mit der Zyklus-Funktion biom-sensor:v2zyklus aktiv.

Ausschalten
* Befehl 2.3 der Befehlsblock löst die Funktion biom-sensor:v2stopp aus.

```
function biom-sensor:v2start

```

↑ 2.2:

```
function biom-sensor:v2zyklus

```

↑ 2.3:

```
function biom-sensor:v2stopp

```

* *Vorrichtungen*
	+ pack.mcmeta
	+ data
		- *biom-sensor*
			* advancements
				+ v2biom\_leere.json
			* functions
				+ v2start.mcfunction
				+ v2zyklus.mcfunction
				+ v2stopp.mcfunction

Variante 2: **Start-Funktion**

Die Start-Funktion gibt dem Spieler einen verzauberten Stock.

```
# Den Spielern wird ein Stock mit Verzauberungsschimmer gegeben.
give @a[distance=..15] minecraft:stick{EigBiomS.2Alle:true,EigBiomS.2Sensor:true,Enchantments:[{id:-1}],display:{Name:'{"text":"Biom-Sensor","color":"dark\_purple","bold":true}',Lore:['"Wähle den Stock aus"','"um nach dem Biom"','"zu testen."'] } }

# Der Chunk wird dauerhaft geladen.
forceload add ~ ~

```

Variante 2: **Zyklus-Funktion**

Die Zyklus-Funktion prüft, ob der Fortschritt für das Biom erhalten wurde und gibt dann eine Schnellleisten-Nachricht aus.

```
# Bei Auswahl des Stockes erhält der Spieler ein temporäres Etikett.
tag @a[nbt={SelectedItem:{tag:{EigBiomS.2Sensor:true} } }] add EtiBiomS.2Ausgewaehlt

# Wenn der Spieler im Biom die Leere ist, erhält er automatisch den Fortschritt dafür und das wird getestet. Nur wenn er diesen hat und den Stock ausgewählt hat, erhält er die Schnellleisten-Nachricht.
title @a[tag=EtiBiomS.2Ausgewaehlt,advancements={biom-sensor:v2biom\_leere=true}] actionbar ["",{"text":"Biom == ","color":"gray","bold":true},{"text":"Die Leere","color":"blue","bold":true}]

# Falls es sich nicht um die Leere handelt wird die Meldung negiert.
title @a[tag=EtiBiomS.2Ausgewaehlt,advancements={biom-sensor:v2biom\_leere=false}] actionbar ["",{"text":"Biom == ","color":"gray","bold":true},{"text":"NICHT Die Leere","color":"red","bold":true}]

# Wenn er den Stock abwählt, wird der Text geleert.
title @a[tag=!EtiBiomS.2Ausgewaehlt,tag=EtiBiomS.2Auswahl] actionbar [""]

# Um die Auswahl und Abwahl zu erfassen wird ihm dazu ein Etikett gegeben oder entfernt.
tag @a[tag=EtiBiomS.2Ausgewaehlt,tag=!EtiBiomS.2Auswahl] add EtiBiomS.2Auswahl
tag @a[tag=!EtiBiomS.2Ausgewaehlt,tag=EtiBiomS.2Auswahl] remove EtiBiomS.2Auswahl

# Wenn Spieler das Biom die Leere betreten haben, wird der Fortschritt wieder entfernt, damit er nur dann vorhanden ist, wenn man aktuell dort ist.
advancement revoke @a[tag=EtiBiomS.2Ausgewaehlt,advancements={biom-sensor:v2biom\_leere=true}] only biom-sensor:v1biom\_leere

# Das temporäre Etikett wird entfernt.
tag @a[tag=EtiBiomS.2Ausgewaehlt] remove EtiBiomS.2Ausgewaehlt

```

```
{
 "criteria": {
 "Biom": {
 "trigger": "minecraft:location",
 "conditions": {
 "location": {
 "biome": "minecraft:the\_void"
 }
 }
 }
 }
}

```

Variante 2: **Stopp-Funktion**

Die Stopp-Funktion löscht den Fortschritt und den Stock wieder.

```
# Der Fortschritt wird entfernt.
advancement revoke @a[advancements={biom-sensor:v2biom\_leere=true}] only biom-sensor:v2biom\_leere

# Das eventuell noch vorhandene Etikett wird entfernt.
tag @a[tag=EtiBiomS.2Auswahl] remove EtiBiomS.2Auswahl

# Der Stock wird entfernt.
clear @a minecraft:stick{EigBiomS.2Alle:true}

# Falls der Stock auf den Boden geworfen wurde, wird er entfernt.
kill @e[type=minecraft:item,nbt={Item:{tag:{EigBiomS.2Alle:true} } }]

# Der Chunk wird entladen.
forceload remove ~ ~

```

## Variante 3

* Grundmodule
* Befehl
* Befehlssignal
* Grund-Element (Befehle)
* Signal-Element (Befehle)
* Signalleitung (Befehle)
 * Prädikate
* JSON-Text
* NBT

* 1 Redstone-Fackel
* 2 Befehlsblock
* 1 Wiederhol-Befehlsblock

Dieser Biom-Sensor erfasst durch ein Prädikat, ob der Spieler sich im Biom die Leere aufhält. Wenn er dabei einen bestimmten Stock in der Schnellleiste ausgewählt hat, erscheint eine Nachricht über der Schnellleiste.

Einschalten
* Befehl 3.1 Auf dem Wiederhol-Befehlsblock befindet sich ein Hebel. Legt man ihn um, wird zuerst der Befehlsblock mit der Start-Funktion biom-sensor:v3start aktiviert.

Wiederholen
* Befehl 3.2 Gleichzeitig wird auch der Wiederhol-Befehlsblock mit der Zyklus-Funktion biom-sensor:v3zyklus aktiv.

Ausschalten
* Befehl 3.3 der Befehlsblock löst die Funktion biom-sensor:v3stopp aus.

```
function biom-sensor:v3start

```

↑ 3.2:

```
function biom-sensor:v3zyklus

```

↑ 3.3:

```
function biom-sensor:v3stopp

```

* *Vorrichtungen*
	+ pack.mcmeta
	+ data
		- *biom-sensor*
			* functions
				+ v3start.mcfunction
				+ v3zyklus.mcfunction
				+ v3stopp.mcfunction
			* predicates
				+ v3biom\_leere.json

Variante 3: **Start-Funktion**

Die Start-Funktion gibt dem Spieler einen Stock, den er auswählen muss um den Sensor zu aktivieren.

```
# Allen Spielern im Umkreis von 15 Blöcken wird jeweils ein Stock mit Verzauberungsschimmer gegeben.
give @a[distance=..15] minecraft:stick{EigBiomS.3Alle:true,EigBiomS.3Sensor:true,Enchantments:[{id:-1}],display:{Name:'{"text":"Biom-Sensor","color":"dark\_purple","bold":true}',Lore:['"Wähle den Stock aus"','"um nach dem Biom"','"zu testen."'] } }

# Der Chunk wird dauerhaft geladen.
forceload add ~ ~

```

Variante 3: **Zyklus-Funktion**

Die Zyklus-Funktion prüft ob der Spieler im entsprechenden Biom sich befindet. Wenn er dabei den Stock ausgewählt hat, erscheint über der Schnellleiste das Biom als Nachricht.

```
# Alle Spieler die den Stock auswählen erhalten ein temporäres Etikett.
tag @a[nbt={SelectedItem:{tag:{EigBiomS.3Sensor:true} } }] add EtiBiomS.3Ausgewaehlt

# Wenn der Spieler in dem Biom die Leere sich befindet und den Stock ausgewählt hat, wird ihm die Schnellleisten-Nachricht angegeben.
title @a[tag=EtiBiomS.3Ausgewaehlt,predicate=biom-sensor:v3biom\_leere] actionbar ["",{"text":"Biom == ","color":"gray","bold":true},{"text":"Die Leere","color":"blue","bold":true}]

# Wenn es sich nicht um das Biom die Leere handelt wird die Meldung negiert.
title @a[tag=EtiBiomS.3Ausgewaehlt,predicate=!biom-sensor:v3biom\_leere] actionbar ["",{"text":"Biom == ","color":"gray","bold":true},{"text":"NICHT Die Leere","color":"red","bold":true}]

# Der Text wird geleert, wenn der Spieler den Stock abwählt.
title @a[tag=!EtiBiomS.3Ausgewaehlt,tag=EtiBiomS.3Auswahl] actionbar [""]

# Um das Auswählen und Abwählen zu erfassen, wird das Etikett vergeben oder entfernt.
tag @a[tag=EtiBiomS.3Ausgewaehlt,tag=!EtiBiomS.3Auswahl] add EtiBiomS.3Auswahl
tag @a[tag=!EtiBiomS.3Ausgewaehlt,tag=EtiBiomS.3Auswahl] remove EtiBiomS.3Auswahl

# Das temporäre Etikett wird entfernt.
tag @a[tag=EtiBiomS.3Ausgewaehlt] remove EtiBiomS.3Ausgewaehlt

```

```
{
 "condition": "minecraft:location\_check",
 "predicate": {
 "biome": "minecraft:the\_void"
 }
}

```

Variante 3: **Stopp-Funktion**

Die Stopp-Funktion löscht den Stock als Drop und als Gegenstand im Inventar wieder.

```
# Das eventuell vorhandene Etikett wird entfernt.
tag @a[tag=EtiBiomS.3Auswahl] remove EtiBiomS.3Auswahl

# Der Stock wird aus dem Inventar aller Spieler entfernt.
clear @a minecraft:stick{EigBiomS.3Alle:true}

# Falls der Stock auf den Boden geworfen wurde, wird er entfernt.
kill @e[type=minecraft:item,nbt={Item:{tag:{EigBiomS.3Alle:true} } }]

# Der Chunk wird entladen.
forceload remove ~ ~

```